
ECS Design
    

Entities and Components
Everything in your application is an Entity, but in ECS you won't work with these
Entities directly - instead you will work with the individual attributes that an Entity
might have.  These attributes are given to an Entity by creating a Component, which holds,
at minimum, the Entity's unique ID, but also can store a value.  For example:
	You're running a 2-dimensional simulation of cars on a highway
	Each car gets its own entity_id e.g. 123
	If the car with ID 123 is blue, we give it a Color Component with value "blue"
	If the same car is moving west at 60mph, we might model this with a Direction Component with value "west" and a Speed Component with value 60
	The car would also have Components such as XCoordinate and YCoordinate to locate it
on the map

Systems
Once your Entities are modeled using Components, you'll create Systems to operate on them.
For example:
	Entities with Speed Components should have their locations regularly updated according to the speed and direction
	We can create a Move System which reads the Speed and Direction Components, calculates how far the car has moved since the last server tick, and updates the Entity's XCoordinate and/or YCoordinate Component accordingly.
	The System will run every tick, only considering Entities which have a Speed Component




  

    
Installation
    

To create an ECSx application, there are a few simple steps:
	install Elixir + erlang/OTP
	install Phoenix (optional)
	create an Elixir/Phoenix project
	fetch ECSx as a dependency for your project
	run the ECSx setup

Elixir and erlang/OTP
If you don't yet have Elixir and erlang/OTP installed on your machine, follow the instructions on the official Installation Page.  
Phoenix
If you plan on hosting your application online, you'll probably want to use Phoenix.  You can skip this step if you only want to run the app locally.  Otherwise, follow the instructions for Phoenix installation (the tutorial project will assume you are using Phoenix).
Create project
If you are using Phoenix, you'll create your new application with the command
$ mix phx.new my_app

Or for a regular Elixir application (with supervision tree):
$ mix new my_app --sup

Install ECSx
To use the ECSx framework in your application, it should be added to the list of dependencies in my_app/mix.exs:
defp deps do
  [
    {:ecsx, "~> 0.3"}
  ]
end
Then (from the root directory of your application) run:
$ mix deps.get

Setup ECSx
With ECSx installed, you can run the setup generator:
$ mix ecsx.setup

which will create the Manager, and two folders to get your project started.
You'll also need to add the Manager to your application's supervision tree:
def start(_type, _args) do
  children = [
    MyApp.Manager
  ]

  opts = [strategy: :one_for_one, name: MyApp.Supervisor]
  Supervisor.start_link(children, opts)
end
You should now have everything you need to start building!



  

    
Initial Setup
    

To demonstrate ECSx in a real-time application, we're going to make a game where each player will control a ship, which can sail around the map, and will attack enemies if they come too close.
Note:  This guide will get you up-and-running with a working game, but it is intentionally generic.  Feel free to experiment with altering details from this implementation to customize your own game.

	First, ensure you have installed Elixir and Phoenix.
	Create the application by running mix phx.new ship
	Add {:ecsx, "~> 0.3"} to your mix.exs deps
	Run mix deps.get
	Run mix ecsx.setup
	Add Ship.Manager to your app's supervision tree




  

    
Backend Basics
    

Defining Component Types
First let's consider the basic properties of a ship:
	Hull Points:  How much damage can it take before it is destroyed
	Armor Rating:  How much is each incoming attack reduced by the ship's defenses
	Attack Damage:  How much damage does its weapon deal to enemies
	Attack Range:  How close must enemies get before the weapon can attack
	Attack Speed:  How much time must you wait in-between attacks
	X Position:  The horizontal position of the ship
	Y Position:  The vertical position of the ship
	X Velocity:  The speed at which the ship is moving, horizontally
	Y Velocity:  The speed at which the ship is moving, vertically

We'll start by creating integer component types for each one of these, except AttackSpeed, which will use float:
$ mix ecsx.gen.component HullPoints integer
$ mix ecsx.gen.component ArmorRating integer
$ mix ecsx.gen.component AttackDamage integer
$ mix ecsx.gen.component AttackRange integer
$ mix ecsx.gen.component XPosition integer
$ mix ecsx.gen.component YPosition integer
$ mix ecsx.gen.component XVelocity integer
$ mix ecsx.gen.component YVelocity integer
$ mix ecsx.gen.component AttackSpeed float

For now, this is all we need to do.  The ECSx generator has automatically set you up with modules for each component type, complete with a simple interface for handling the components.  We'll see this in action soon.
Our First System
Having set up the component types which will model our game data, let's think about the Systems which will organize game logic.  What makes our game work?
	Ships change position based on velocity
	Ships target other ships for attack when they are within range
	Ships with valid targets should attack the target, reducing its hull points
	Ships with zero or less hull points are destroyed
	Players change the velocity of their ship using an input device
	Players can see a display of the area around their ship

Let's start with changing position based on velocity.  We'll call it Driver:
$ mix ecsx.gen.system Driver

Head over to the generated file lib/ship/systems/driver.ex and we'll add some code:
defmodule Ship.Systems.Driver do
  ...
  use ECSx.System

  alias Ship.Components.XPosition
  alias Ship.Components.YPosition
  alias Ship.Components.XVelocity
  alias Ship.Components.YVelocity

  def run do
    for {entity, x_velocity} <- XVelocity.get_all() do
      x_position = XPosition.get_one(entity)
      new_x_position = x_position + x_velocity
      # By default, an entity can only have one component of each type.  
      # Adding a second will overwrite the first.
      XPosition.add(entity, new_x_position)
    end

    # Once the x-values are updated, do the same for the y-values
    for {entity, y_velocity} <- YVelocity.get_all() do
      y_position = YPosition.get_one(entity)
      new_y_position = y_position + y_velocity
      YPosition.add(entity, new_y_position)
    end

    # run/0 should always return :ok
    :ok
  end
end
Now whenever a ship gains velocity, this system will update the position accordingly over time.  Keep in mind that the velocity is relative to the server's tick rate, which by default is 20.  This means the unit of measurement is "game units per 1/20th of a second".
For example, if you want the speed to move from XPosition 0 to XPosition 100 in one second, you divide the distance 100 by the tick rate 20, to see that an XVelocity of 5 is appropriate.  The tick rate can be changed in config/config.ex.
Targeting & Attacking
Next let's move on to a more complicated part of the game - attacking.  We'll start by considering the conditions which must be met in order to attack a given target:
	Target must be a ship
	Target must be within your ship's attack range
	You must not have attacked too recently (based on attack speed)

For each of these conditions, we want to use the presence or absence of a component as the signal to a system that action is to be taken.  For example, in the Driver system, these were the Velocity components - for each Velocity component, we made a Position update.  
First, for determining whether a given entity is a ship, we will simply use the existing HullPoints component, because only ships will have HullPoints.
Second, for confirming the attack range, we'll make a new component type SeekingTarget which will signal to a Targeting system that a ship's proximity to other ships must be continuously calculated until a valid target is found.  Then another new component type AttackTarget will replace SeekingTarget, signaling to the Targeting system that we no longer need to check for new targets.  Instead, an Attacking system will detect the AttackTarget and handle the final step of the attacking process.
The final attack requirement is that after a successful attack, the ship's weapon must wait for a cooldown period, based on the attack speed.  To model this cooldown period, we will create an AttackCooldown component type, which will store the time at which the cooldown expires.
With this plan in place, let's go ahead and create the component types, starting with SeekingTarget.  Since the presence of this component alone fulfills its purpose, without the need to store additional data, this is the appropriate use-case for a Tag:
$ mix ecsx.gen.tag SeekingTarget

Once a target is found, the AttackTarget component will be needed, and this time a Tag will not be enough, because we need to store the ID of the target.  Likewise with AttackCooldown, which must store the timestamp of the cooldown's expiration.
$ mix ecsx.gen.component AttackTarget binary
$ mix ecsx.gen.component AttackCooldown datetime

Note:  In our case, we're using binary IDs to represent Entities, and Elixir DateTime structs for cooldown expirations.  If you're planning on using different types, such as integer IDs for entities, or storing timestamps as integers, simply adjust the parameters accordingly.

Before we set up the systems, let's make a helper module for storing any shared mathematical logic.  In particular, we'll need a function for calculating the distance between two entities.  This will come in handy for several systems in the future.
defmodule Ship.SystemUtils do
  @moduledoc """
  Useful math functions used by multiple systems.
  """

  alias Ship.Components.XPosition
  alias Ship.Components.YPosition

  def distance_between(entity_1, entity_2) do
    x_1 = XPosition.get_one(entity_1)
    x_2 = XPosition.get_one(entity_2)
    y_1 = YPosition.get_one(entity_1)
    y_2 = YPosition.get_one(entity_2)

    x = abs(x_1 - x_2)
    y = abs(y_1 - y_2)

    :math.sqrt(x ** 2 + y ** 2)
  end
end
Now we're onto the Targeting system, which operates only on entities with the SeekingTarget component, checking the distance to all other ships, and comparing them to the entity's attack range.  When an enemy ship is found to be within range, we can remove SeekingTarget and replace it with an AttackTarget:
$ mix ecsx.gen.system Targeting

defmodule Ship.Systems.Targeting do
  ...
  use ECSx.System

  alias Ship.Components.AttackRange
  alias Ship.Components.AttackTarget
  alias Ship.Components.HullPoints
  alias Ship.Components.SeekingTarget
  alias Ship.SystemUtils

  def run do
    entities = SeekingTarget.get_all()

    Enum.each(entities, &attempt_target/1)
  end

  defp attempt_target(self) do
    case look_for_target(self) do
      nil -> :noop
      {target, _hp} -> add_target(self, target)
    end
  end 

  defp look_for_target(self) do
    # For now, we're assuming anything which has HullPoints can be attacked
    HullPoints.get_all()
    # ... except your own ship!
    |> Enum.reject(fn {possible_target, _hp} -> possible_target == self end)
    |> Enum.find(fn {possible_target, _hp} ->
      distance_between = SystemUtils.distance_between(possible_target, self)
      range = AttackRange.get_one(self)

      distance_between < range
    end)
  end

  defp add_target(self, target) do
    SeekingTarget.remove(self)
    AttackTarget.add(self, target)
  end
end
The Attacking system will also check distance, but only to the target ship, in case it has moved out-of-range.  If not, we just need to check on the cooldown, and do the attack.
$ mix ecsx.gen.system Attacking

defmodule Ship.Systems.Attacking do
  ...
  use ECSx.System

  alias Ship.Components.ArmorRating
  alias Ship.Components.AttackCooldown
  alias Ship.Components.AttackDamage
  alias Ship.Components.AttackRange
  alias Ship.Components.AttackSpeed
  alias Ship.Components.AttackTarget
  alias Ship.Components.HullPoints
  alias Ship.Components.SeekingTarget
  alias Ship.SystemUtils
  
  def run do
    attack_targets = AttackTarget.get_all()

    Enum.each(attack_targets, &attack_if_ready/1)
  end

  defp attack_if_ready({self, target}) do
    cond do
      SystemUtils.distance_between(self, target) > AttackRange.get_one(self) ->
        # If the target ever leaves our attack range, we want to remove the AttackTarget
        # and begin searching for a new one.
        AttackTarget.remove(self)
        SeekingTarget.add(self)

      AttackCooldown.exists?(self) ->
        # We're still within range, but waiting on the cooldown
        :noop

      :otherwise ->
        deal_damage(self, target)
        add_cooldown(self)
    end
  end

  defp deal_damage(self, target) do
    attack_damage = AttackDamage.get_one(self)
    # Assuming one armor rating always equals one damage
    reduction_from_armor = ArmorRating.get_one(target)
    final_damage_amount = attack_damage - reduction_from_armor

    target_current_hp = HullPoints.get_one(target)
    target_new_hp = target_current_hp - final_damage_amount

    HullPoints.add(target, target_new_hp)
  end

  defp add_cooldown(self) do
    now = DateTime.utc_now()
    ms_between_attacks = calculate_cooldown_time(self)
    cooldown_until = DateTime.add(now, ms_between_attacks, :millisecond)

    AttackCooldown.add(self, cooldown_until)
  end

  # We're going to model AttackSpeed with a float representing attacks per second.
  # The goal here is to convert that into milliseconds per attack.
  defp calculate_cooldown_time(self) do
    attacks_per_second = AttackSpeed.get_one(self)
    seconds_per_attack = 1 / attacks_per_second

    ceil(seconds_per_attack * 1000)
  end
end
Phew, that was a lot!  But we're still using the same basic concepts:  get_all/0 to fetch the list of all relevant entities, then get_one/1 and exists?/1 to check specific attributes of the entities, and finally add/2 for creating new components, or overwriting existing ones.  We're also starting to see the use of remove/1 for excluding an entity from game logic which is no longer necessary.
Cooldowns
Our attacking system will add a cooldown with an expiration timestamp, but the next step is to ensure the cooldown component is removed from the entity once the time is reached, so it can attack again.  For that, we'll create a CooldownExpiration system:
$ mix ecsx.gen.system CooldownExpiration

defmodule Ship.Systems.CooldownExpiration do
  ...
  use ECSx.System

  alias Ship.Components.AttackCooldown

  def run do
    now = DateTime.utc_now()
    cooldowns = AttackCooldown.get_all()
      
    Enum.each(cooldowns, &remove_when_expired(&1, now))
  end

  defp remove_when_expired({entity, timestamp}, now) do
    case DateTime.compare(now, timestamp) do
      :lt -> :noop
      _ -> AttackCooldown.remove(entity)
    end
  end
end
This system will check the cooldowns on each game tick, removing them as soon as the expiration time is reached.
Death & Destruction
Next let's handle what happens when a ship has its HP reduced to zero or less:
$ mix ecsx.gen.component DestroyedAt datetime

$ mix ecsx.gen.system Destruction

defmodule Ship.Systems.Destruction do
  ...
  use ECSx.System

  alias Ship.Components.ArmorRating
  alias Ship.Components.AttackCooldown
  alias Ship.Components.AttackDamage
  alias Ship.Components.AttackRange
  alias Ship.Components.AttackSpeed
  alias Ship.Components.AttackTarget
  alias Ship.Components.DestroyedAt
  alias Ship.Components.HullPoints
  alias Ship.Components.SeekingTarget
  alias Ship.Components.XPosition
  alias Ship.Components.XVelocity
  alias Ship.Components.YPosition
  alias Ship.Components.YVelocity

  def run do
    ships = HullPoints.get_all()

    Enum.each(ships, fn {entity, hp} ->
      if hp <= 0, do: destroy(entity)
    end)
  end

  defp destroy(entity) do
    ArmorRating.remove(entity)
    AttackCooldown.remove(entity)
    AttackDamage.remove(entity)
    AttackRange.remove(entity)
    AttackSpeed.remove(entity)
    AttackTarget.remove(entity)
    HullPoints.remove(entity)
    SeekingTarget.remove(entity)
    XPosition.remove(entity)
    XVelocity.remove(entity)
    YPosition.remove(entity)
    YVelocity.remove(entity)

    # when a ship is destroyed, other ships should stop targeting it
    untarget(entity)

    DestroyedAt.add(entity, DateTime.utc_now())
  end

  defp untarget(target) do
    for entity <- AttackTarget.search(target) do
      AttackTarget.remove(entity)
      SeekingTarget.add(entity)
    end
  end
end
In this example we remove all the components the entity might have, then add a new DestroyedAt component with the current timestamp.  If we wanted some components to persist - such as the position and/or velocity, so the wreckage could still be visible on the player displays - we could keep them around and possibly have another system clean them up later on.  Likewise if there were other components to add, such as a RespawnTimer or FinalScore, we could add them here as well.
Initializing Components
By now you might be wondering "How did those components get created in the first place?"  We have code for adding AttackCooldown and DestroyedAt, when needed, but the basic components for the ships still need to be added before the game can even start.  For that, we'll check out lib/ship/manager.ex:
defmodule Ship.Manager do
  ...
  use ECSx.Manager, tick_rate: 20

  setup do
    # Load your initial components
  end

  def components do
    ...
  end

  def systems do
    ...
  end
end
This module holds four critical pieces of data - the server's tick rate, data initialization, a list of every valid component type, and a list of each game system in the order they are to be run.  Let's initialize some ship data inside the setup block:
setup do
  for _ships <- 1..40 do
    # First generate a unique ID to represent the new entity
    entity = Ecto.UUID.generate()

    # Then use that ID to create the components which make up a ship    
    Ship.Components.ArmorRating.add(entity, 0)
    Ship.Components.AttackDamage.add(entity, 5)
    Ship.Components.AttackRange.add(entity, 10)
    Ship.Components.AttackSpeed.add(entity, 1.05)
    Ship.Components.HullPoints.add(entity, 50)
    Ship.Components.SeekingTarget.add(entity)
    Ship.Components.XPosition.add(entity, Enum.random(1..100))
    Ship.Components.YPosition.add(entity, Enum.random(1..100))
  end
end
Now when the server starts, there will be forty ships set up and ready to go.



  

    
Web Frontend with LiveView
    

Since we're using Phoenix, we can take advantage of the many features it brings for building a web interface.
Player Auth
When it comes to player auth, there are two sides to the coin:  authentication (AuthN) and authorization (AuthZ).  The former refers to verifying the identity of a player (and will be our primary focus, for now), while the latter refers to checking whether a user has permission to take a restricted action.
Phoenix comes with an AuthN generator built-in, which should be more than enough for our needs:
$ mix phx.gen.auth Players Player players --binary-id
$ mix deps.get
$ mix ecto.migrate

This will expect players to register an email and password, which will be used to log in.  A unique ID will also be created for each player upon registration, allowing us to begin thinking of players as entities.  However, we can't just take the player input and start creating components with it - only systems can create components.  Instead, we'll use a special component type provided for this purpose: ECSx.ClientEvents.
Client Input via LiveView
First consider the goals for our frontend:
	Authenticate the player and hold player ID
	Spawn the player's ship upon connection (writes components)
	Hold the coordinates for the player's ship
	Hold the coordinates for enemy ships
	Validate user input to move the ship (writes components)

When we need to write components, ECSx.ClientEvents will be our line of communication from the frontend to the backend.  
Let's create /lib/ship_web/game_live.ex and put it to use:
defmodule ShipWeb.GameLive do
  use ShipWeb, :live_view

  alias Ship.Components.HullPoints
  alias Ship.Components.XPosition
  alias Ship.Components.YPosition

  def mount(_params, %{"player_token" => token} = _session, socket) do
    # This context function was generated by phx.gen.auth
    player = Ship.Players.get_player_by_session_token(token)
    
    # We must spawn the ship before processing any other input
    ECSx.ClientEvents.add(player.id, :spawn_ship)

    socket =
      socket
      |> assign(player_entity: player.id)
      # Keeping a set of currently held keys will allow us to prevent duplicate keydown events
      |> assign(keys: MapSet.new())
      # We don't know where the ship will spawn, yet
      |> assign(x_coord: nil, y_coord: nil, current_hp: nil)

    :timer.send_interval(50, :load_player_info)

    {:ok, socket}
  end

  def handle_info(:load_player_info, socket) do
    # This will run every 50ms to keep the client assigns updated
    x = XPosition.get_one(socket.assigns.player_entity)
    y = YPosition.get_one(socket.assigns.player_entity)
    hp = HullPoints.get_one(socket.assigns.player_entity)

    {:noreply, assign(socket, x_coord: x, y_coord: y, current_hp: hp)}
  end

  def handle_event("keydown", %{"key" => key}, socket) do
    if MapSet.member?(socket.assigns.keys, key) do
      # Already holding this key - do nothing
      {:noreply, socket}
    else
      # We only want to add a client event if the key is defined by the `keydown/1` helper below
      maybe_add_client_event(socket.assigns.player_entity, key, &keydown/1)
      {:noreply, assign(socket, keys: MapSet.put(socket.assigns.keys, key))}
    end
  end

  def handle_event("keyup", %{"key" => key}, socket) do
    # We don't have to worry about duplicate keyup events
    # But once again, we will only add client events for keys that actually do something
    maybe_add_client_event(socket.assigns.player_entity, key, &keyup/1)
    {:noreply, assign(socket, keys: MapSet.delete(socket.assigns.keys, key))}
  end

  defp maybe_add_client_event(player_entity, key, fun) do
    case fun.(key) do
      :noop -> :ok
      event -> ECSx.ClientEvents.add(player_entity, event)
    end
  end

  defp keydown(key) when key in ~w(w W ArrowUp), do: {:move, :north}
  defp keydown(key) when key in ~w(a A ArrowLeft), do: {:move, :west}
  defp keydown(key) when key in ~w(s S ArrowDown), do: {:move, :south}
  defp keydown(key) when key in ~w(d D ArrowRight), do: {:move, :east}
  defp keydown(_key), do: :noop

  defp keyup(key) when key in ~w(w W ArrowUp), do: {:stop_move, :north}
  defp keyup(key) when key in ~w(a A ArrowLeft), do: {:stop_move, :west}
  defp keyup(key) when key in ~w(s S ArrowDown), do: {:stop_move, :south}
  defp keyup(key) when key in ~w(d D ArrowRight), do: {:stop_move, :east}
  defp keyup(_key), do: :noop

  def render(assigns) do
    ~H"""
    <div id="game" phx-window-keydown="keydown" phx-window-keyup="keyup">
      <p>Player ID: <%= @player_entity %></p>
      <p>Player Coords: <%= inspect({@x_coord, @y_coord}) %></p>
      <p>Hull Points: <%= @current_hp %></p>
    </div>
    """
  end
end
Handling Client Events
Finally, spin up a new system for handling the events:
$ mix ecsx.gen.system ClientEventHandler

defmodule Ship.Systems.ClientEventHandler do
  ...
  use ECSx.System

  alias Ship.Components.ArmorRating
  alias Ship.Components.AttackDamage
  alias Ship.Components.AttackRange
  alias Ship.Components.AttackSpeed
  alias Ship.Components.HullPoints
  alias Ship.Components.SeekingTarget
  alias Ship.Components.XPosition
  alias Ship.Components.XVelocity
  alias Ship.Components.YPosition
  alias Ship.Components.YVelocity

  def run do
    client_events = ECSx.ClientEvents.get_and_clear()

    Enum.each(client_events, &process_one/1)
  end

  defp process_one(player, :spawn_ship) do
    # We'll give player ships better stats than the enemy ships
    # (otherwise the game would be very short!)
    ArmorRating.add(player, 2)
    AttackDamage.add(player, 6)
    AttackRange.add(player, 15)
    AttackSpeed.add(player, 1.2)
    HullPoints.add(player, 75)
    SeekingTarget.add(player)
    XPosition.add(player, Enum.random(1..100))
    YPosition.add(player, Enum.random(1..100))
  end

  # Note Y movement will use screen position (increasing Y goes south)
  defp process_one({player, {:move, :north}}), do: YVelocity.add(player, -1)
  defp process_one({player, {:move, :south}}), do: YVelocity.add(player, 1)
  defp process_one({player, {:move, :east}}), do: XVelocity.add(player, 1)
  defp process_one({player, {:move, :west}}), do: XVelocity.add(player, -1)

  defp process_one({player, {:stop_move, :north}}), do: YVelocity.remove_one(player, -1)
  defp process_one({player, {:stop_move, :south}}), do: YVelocity.remove_one(player, 1)
  defp process_one({player, {:stop_move, :east}}), do: XVelocity.remove_one(player, 1)
  defp process_one({player, {:stop_move, :west}}), do: XVelocity.remove_one(player, -1)
end
Notice how the LiveView client can write to ECSx.ClientEvents, while the system handles and also clears the events.  This ensures that we don't process the same event twice, nor will any events get "lost" and not processed.
Creating a Phoenix Route
Head into router.ex and look for the new scope which uses :require_authenticated_player.  We're going to add a new route for our game interface:
scope "/", ShipWeb do
  pipe_through [:browser, :require_authenticated_player]

  live "/game", ShipWeb.GameLive
  ...
end
Now we can run
$ iex -S mix phx.server

and go to localhost:4000/game to test the input.  Once you are logged in, wait for the player coords to display (this will be the indicator that your ship has spawned), and try moving around with WASD or arrow keys!
Loading Screen
You might notice that while the ship is spawning, the Player Coords and Hull Points don't display properly - this isn't a major issue now, but once our coordinates are being used by a more sophisticated display, this will not be acceptable.  What we need is a loading screen to show the user until the necessary data is properly loaded.
First, let's create a new ECSx.Tag to mark when a player's ship has finished spawning:
$ mix ecsx.gen.tag PlayerSpawned

Then we'll add this tag at the end of the :spawn_ship client event
defmodule Ship.Systems.ClientEventHandler do
  ...
  alias Ship.Components.PlayerSpawned
  ...
  defp process_one(player, :spawn_ship) do
    ...
    PlayerSpawned.add(player)
  end
  ...
end
Now we'll update our LiveView to use a new @loading assign which is initially set to true, then set to false after the ship is spawned and the data is loaded for the first time.
defmodule ShipWeb.GameLive do
  ...
  alias Ship.Components.PlayerSpawned
  ...
  def mount(_params, %{"player_token" => token} = _session, socket) do
    player = Ship.Players.get_player_by_session_token(token)
    ECSx.ClientEvents.add(player.id, :spawn_ship)

    socket =
      socket
      |> assign(player_entity: player.id)
      |> assign(keys: MapSet.new())
      # This gets its own helper in case we need to return to this state again later
      |> assign_loading_state()

    # The first load will now have additional responsibilities
    send(self(), :first_load)

    {:ok, socket}
  end

  defp assign_loading_state(socket) do
    assign(socket,
      x_coord: nil,
      y_coord: nil,
      current_hp: nil,
      # This new assign will control whether the loading screen is shown
      loading: true
    )
  end

  def handle_info(:first_load, socket) do
    # Don't start fetching components until after spawn is complete!
    :ok = wait_for_spawn(socket.assigns.player_entity)

    socket =
      socket
      |> assign_player_ship()
      |> assign(loading: false)

    # We want to keep up-to-date on this info
    :timer.send_interval(50, :refresh)

    {:noreply, socket}
  end

  def handle_info(:refresh, socket) do
    {:noreply, assign_player_ship(socket)}
  end

  defp wait_for_spawn(player_entity) do
    if PlayerSpawned.exists?(player_entity) do
      :ok
    else
      Process.sleep(10)
      wait_for_spawn(player_entity)
    end
  end

  # Our previous :load_player_info handler becomes a shared helper for the new handlers
  defp assign_player_ship(socket) do
    x = XPosition.get_one(socket.assigns.player_entity)
    y = YPosition.get_one(socket.assigns.player_entity)
    hp = HullPoints.get_one(socket.assigns.player_entity)

    assign(socket, x_coord: x, y_coord: y, current_hp: hp)
  end

  ...

  def render(assigns) do
    ~H"""
    <div id="game" phx-window-keydown="keydown" phx-window-keyup="keyup">
      <%= if @loading do %>
        <p>Loading...</p>
      <% else %>
        <p>Player ID: <%= @player_entity %></p>
        <p>Player Coords: <%= inspect({@x_coord, @y_coord}) %></p>
        <p>Hull Points: <%= @current_hp %></p>
      <% end %>
    </div>
    """
  end
end
Player GUI using SVG
One of the simplest ways to build a display for web is with SVG.  Each entity can be represented by a single SVG element, which only requires its coordinates.  Then a viewBox can zoom the player's display in to show just the local area around their ship.
defmodule ShipWeb.GameLive do
  ...
  def render(assigns) do
    ~H"""
    <div id="game" phx-window-keydown="keydown" phx-window-keyup="keyup">
      <svg
        viewBox={"#{@x_offset} #{@y_offset} #{@screen_width} #{@screen_height}"}
        preserveAspectRatio="xMinYMin slice"
      >
        <rect width={@game_world_size} height={@game_world_size} fill="#72eff8" />

        <%= if @loading do %>
          <text x={div(@screen_width, 2)} y={div(@screen_height, 2)} style="font: 1px serif">
            Loading...
          </text>
        <% else %>
          <image
            x={@x_coord}
            y={@y_coord}
            width="1"
            height="1"
            href={Routes.static_path(@socket, "/images/" <> @player_ship_image_file)}
          />
          <%= for {_entity, x, y, image_file} <- @other_ships do %>
            <image
              x={x}
              y={y}
              width="1"
              height="1"
              href={Routes.static_path(@socket, "/images/" <> image_file)}
            />
          <% end %>
          <text x={@x_offset} y={@y_offset + 1} style="font: 1px serif">
            Hull Points: <%= @current_hp %>
          </text>
        <% end %>
      </svg>
    </div>
    """
  end
end
We've added a lot here, so let's go line-by-line:
We're filling the screen with an svg viewBox, which takes four arguments.  The first two - x and y offsets - tell the viewBox what area of the map to focus on, while the latter two - screen width and height - tell it how much to zoom in.  To get the offsets, we'll need to calculate the coordinate pair which should be at the very top-left of the player's view.  This will need to be updated every time the player moves.  The screen width and height (measured by game coordinates) will be assigned on mount and won't change.  The preserveAspectRatio has two parts: xMinYMin means we define the offset by the top-left coordinate, and slice means we're only expecting to show a "slice" of the map in our viewBox, not the whole thing.
Our first element will be a simple rect (rectangle) with a light-blue fill to cover the entire world map.  This will be the "background" - representing the ocean.  The game world will consistently be 100x100, so we can assign a game_world_size of 100 for this purpose.
The loading screen will still use the same viewBox and background, but with only one other element - text to display a loading message in the center of the screen.  One curiosity regarding the font size:  we use 1px, but you'll see later that the text is actually quite large.  This is because the viewBox considers each game "tile" to be 1 pixel, and automatically scales these pixels up to a larger size based on the screen width and height compared to the size of the browser window.  So when we say font: 1px it means the text will be as tall as one game tile.
Once the game is finished loading, we'll display three things:  the player's ship, other ships, and the player's current HP.
For the player's ship, we'll make an image element, using the existing x and y coordinates, defining the size as one game tile, and pointing to the player's ship image file.
For other ships, we'll need a new assign to hold that data - ID and coordinates, at minimum.  Then each one will get an image just like the player's ship.
Lastly, we'll put an HP display near the top-left corner.
Next is to update our LiveView with these new assigns:
defmodule ShipWeb.GameLive do
  ...
  alias Ship.Components.ImageFile
  ...
  def mount(_params, %{"player_token" => token} = _session, socket) do
    player = Ship.Players.get_player_by_session_token(token)
    ECSx.ClientEvents.add(player.id, :spawn_ship)

    socket =
      socket
      |> assign(player_entity: player.id)
      |> assign(keys: MapSet.new())
      # These will configure the scale of our display compared to the game world
      |> assign(game_world_size: 100, screen_height: 30, screen_width: 50)
      |> assign_loading_state()
    
    send(self(), :first_load)

    {:ok, socket}
  end

  defp assign_loading_state(socket) do
    assign(socket,
      x_coord: nil,
      y_coord: nil,
      current_hp: nil,
      player_ship_image_file: nil,
      other_ships: [],
      x_offset: 0,
      y_offset: 0,
      loading: true
    )
  end

  def handle_info(:first_load, socket) do
    :ok = wait_for_spawn(socket.assigns.player_entity)

    socket =
      socket
      |> assign_player_ship()
      |> assign_other_ships()
      |> assign_offsets()
      |> assign(loading: false)

    :timer.send_interval(50, :refresh)

    {:noreply, socket}
  end

  def handle_info(:refresh, socket) do
    socket =
      socket
      |> assign_player_ship()
      |> assign_other_ships()
      |> assign_offsets()

    {:noreply, socket}
  end
  ...
  defp assign_player_ship(socket) do
    x = XPosition.get_one(socket.assigns.player_entity)
    y = YPosition.get_one(socket.assigns.player_entity)
    hp = HullPoints.get_one(socket.assigns.player_entity)
    image = ImageFile.get_one(socket.assigns.player_entity)

    assign(socket, x_coord: x, y_coord: y, current_hp: hp, player_ship_image_file: image)
  end

  defp assign_other_ships(socket) do
    other_ships =
      Enum.reject(all_ships(), fn {entity, _, _, _} -> entity == socket.assigns.player_entity end)

    assign(socket, other_ships: other_ships)
  end

  defp all_ships do
    for {ship, _hp} <- HullPoints.get_all() do
      x = XPosition.get_one(ship)
      y = YPosition.get_one(ship)
      image = ImageFile.get_one(ship)
      {ship, x, y, image}
    end
  end

  defp assign_offsets(socket) do
    # Note: the socket must already have updated player coordinates before assigning offsets!
    %{screen_width: screen_width, screen_height: screen_height} = socket.assigns
    %{x_coord: x, y_coord: y, game_world_size: game_world_size} = socket.assigns

    x_offset = calculate_offset(x, screen_width, game_world_size)
    y_offset = calculate_offset(y, screen_height, game_world_size)

    assign(socket, x_offset: x_offset, y_offset: y_offset)
  end

  defp calculate_offset(coord, screen_size, game_world_size) do
    case coord - div(screen_size, 2) do
      offset when offset < 0 -> 0
      offset when offset > game_world_size - screen_size -> game_world_size - screen_size
      offset -> offset
    end
  end
end
Next let's create the ImageFile components when a ship is spawned:
defmodule Ship.Manager do
  ...
  setup do
    for _ships <- 1..40 do
      ...
      Ship.Components.ImageFile.add(entity, "npc_ship.svg")
    end
  end
  ...
end
defmodule Ship.Systems.ClientEventHandler do
  ...
  alias Ship.Components.ImageFile
  ...
  defp process_one({player, :spawn_ship}) do
    ...
    ImageFile.add(player, "player_ship.svg")
    PlayerSpawned.add(player)
  end
  ...
end
Lastly, we'll need the player_ship.svg and npc_ship.svg files.  Right-click on the links and save them to priv/static/images/, where they will be found by our Routes.static_path/2 calls in the LiveView template.
Now running
$ iex -S mix phx.server

and heading to localhost:4000/game should provide a usable game interface to move your ship around, ideally keeping it out of attack range of enemy ships, while remaining close enough for your own ship to attack (remember that we gave the player ship a longer attack range than the enemy ships).
Projectile Animations
Currently the most challenging part of the game is knowing when your ship is attacking, and when it is being attacked.  Let's implement a new feature to make attacks visible to the player(s).  There are several ways to go about this;  we're going to take an approach that showcases ECS design:
	Instead of an attack immediately dealing damage, it will spawn a cannonball entity
	The cannonball entity will have position and velocity components, like ships do
	It will also have new components such as ProjectileTarget and ProjectileDamage
	A Projectile system will guide it to its target, then destroy the cannonball and deal damage
	In our LiveView, we'll create a new assign to hold the locations of projectiles
	The new assign will be used to create SVG elements
	To help fetch locations for projectiles only, we'll add an IsProjectile tag

Start by running the generator commands for our new components, systems, and tag:
$ mix ecsx.gen.component ProjectileTarget binary
$ mix ecsx.gen.component ProjectileDamage integer
$ mix ecsx.gen.system Projectile
$ mix ecsx.gen.tag IsProjectile

Then we need to update the Attacking system to spawn projectiles instead of immediately dealing damage.  We'll replace the existing deal_damage/2 with a spawn_projectile/2:
defmodule Ship.Systems.Attacking do
  ...
  use ECSx.System

  alias Ship.Components.AttackCooldown
  alias Ship.Components.AttackDamage
  alias Ship.Components.AttackRange
  alias Ship.Components.AttackSpeed
  alias Ship.Components.AttackTarget
  alias Ship.Components.ImageFile
  alias Ship.Components.IsProjectile
  alias Ship.Components.ProjectileDamage
  alias Ship.Components.ProjectileTarget
  alias Ship.Components.SeekingTarget
  alias Ship.Components.XPosition
  alias Ship.Components.YPosition
  alias Ship.SystemUtils
  ...
  defp attack_if_ready({self, target}) do
    cond do
      ...
      :otherwise ->
        spawn_projectile(self, target)
        add_cooldown(self)
    end
  end

  defp spawn_projectile(self, target) do
    attack_damage = AttackDamage.get_one(self)
    x = XPosition.get_one(self)
    y = YPosition.get_one(self)
    # Armor reduction should wait until impact to be calculated
    cannonball_entity = Ecto.UUID.generate()

    IsProjectile.add(cannonball_entity)
    XPosition.add(cannonball_entity, x)
    YPosition.add(cannonball_entity, y)
    ImageFile.add(cannonball_entity, "cannonball.svg")
    ProjectileTarget.add(cannonball_entity, target)
    ProjectileDamage.add(cannonball_entity, attack_damage)
  end
  ...
end
Notice we didn't worry about setting the velocity, because that will be handled by the Projectile system:
defmodule Ship.Systems.Projectile do
  ...
  use ECSx.System

  alias Ship.Components.ArmorRating
  alias Ship.Components.HullPoints
  alias Ship.Components.IsProjectile
  alias Ship.Components.ProjectileDamage
  alias Ship.Components.ProjectileTarget
  alias Ship.Components.XPosition
  alias Ship.Components.XVelocity
  alias Ship.Components.YPosition
  alias Ship.Components.YVelocity

  @cannonball_speed 3

  def run do
    projectiles = IsProjectile.get_all()
      
    Enum.each(projectiles, fn projectile ->
      case ProjectileTarget.get_one(projectile) do
        nil ->
          # The target has already been destroyed
          destroy_projectile(projectile)

        target ->
          continue_seeking_target(projectile, target)
      end
    end)
  end

  defp continue_seeking_target(projectile, target) do
    {dx, dy, distance} = get_distance_to_target(projectile, target)

    case distance do
      0 ->
        collision(projectile, target)

      distance when distance / @cannonball_speed <= 1 ->
        move_directly_to_target(projectile, {dx, dy})

      distance ->
        adjust_velocity_towards_target(projectile, {distance, dx, dy})
    end
  end

  defp get_distance_to_target(projectile, target) do
    target_x = XPosition.get_one(target)
    target_y = YPosition.get_one(target)
    target_dx = XVelocity.get_one(target) || 0
    target_dy = YVelocity.get_one(target) || 0
    target_next_x = target_x + target_dx
    target_next_y = target_y + target_dy

    x = XPosition.get_one(projectile)
    y = YPosition.get_one(projectile)

    dx = target_next_x - x
    dy = target_next_y - y

    {dx, dy, ceil(:math.sqrt(dx ** 2 + dy ** 2))}
  end

  defp collision(projectile, target) do
    damage_target(projectile, target)
    destroy_projectile(projectile)
  end

  defp damage_target(projectile, target) do
    damage = ProjectileDamage.get_one(projectile)
    reduction_from_armor = ArmorRating.get_one(target)
    final_damage_amount = amount - reduction_from_armor

    target_current_hp = HullPoints.get_one(target)
    target_new_hp = target_current_hp - final_damage_amount

    HullPoints.add(target, target_new_hp)
  end

  defp destroy_projectile(projectile) do
    IsProjectile.remove(projectile)
    XPosition.remove(projectile)
    YPosition.remove(projectile)
    XVelocity.remove(projectile)
    YVelocity.remove(projectile)
    ImageFile.remove(projectile)
    ProjectileTarget.remove(projectile)
    ProjectileDamage.remove(projectile)
  end

  defp move_directly_to_target(projectile, {dx, dy}) do
    XVelocity.add(projectile, dx)
    YVelocity.add(projectile, dy)
  end

  defp adjust_velocity_towards_target(projectile, {distance, dx, dy}) do
    # We know what is needed, but we need to slow it down, so its travel
    # will take more than one tick.  Otherwise the player will not see it!
    ticks_away = ceil(distance / @cannonball_speed)
    adjusted_dx = div(dx, ticks_away)
    adjusted_dy = div(dy, ticks_away)

    XVelocity.add(projectile, adjusted_dx)
    YVelocity.add(projectile, adjusted_dy)
  end
end
Note that we rely on the absence of a ProjectileTarget to know that the target is already destroyed.  Currently our Destruction system does have an untarget feature for removing target components upon destruction, but this only applies to AttackTargets.  We'll want to expand this feature to also cover ProjectileTargets:
defmodule Ship.Systems.Destruction do
  ...
  alias Ship.Components.ProjectileTarget
  ...
  defp untarget(target) do
    for ship <- AttackTarget.search(target) do
      AttackTarget.remove(ship)
      SeekingTarget.add(ship)
    end

    for projectile <- ProjectileTarget.search(target) do
      ProjectileTarget.remove(projectile)
    end
  end
end
Our final task is to render these projectiles in the LiveView.  Let's start by adding a new assign:
defmodule Ship.GameLive do
  ...
  alias Ship.Components.IsProjectile
  ...
  defp assign_loading_state(socket) do
    assign(socket,
      ...
      projectiles: []
    )
  end

  def handle_info(:first_load, socket) do
    ...
    socket =
      socket
      |> assign_player_ship()
      |> assign_other_ships()
      |> assign_projectiles()
      |> assign_offsets()
      |> assign(loading: false)
    ...
  end

  def handle_info(:refresh, socket) do
    socket =
      socket
      |> assign_player_ship()
      |> assign_other_ships()
      |> assign_projectiles()
      |> assign_offsets()
    ...
  end
  ...
  defp assign_projectiles(socket) do
    projectiles =
      for projectile <- IsProjectile.get_all() do
        x = XPosition.get_one(projectile)
        y = YPosition.get_one(projectile)
        image = ImageFile.get_one(projectile)
        {projectile, x, y, image}
      end

    assign(socket, projectiles: projectiles)
  end
  ...
end
Then we'll update the render to include the projectiles:
defmodule Ship.GameLive do
  def render(assigns) do
    ~H"""
    ...
    <%= for {_entity, x, y, image_file} <- @projectiles do %>
      <image
        x={x}
        y={y}
        width="1"
        height="1"
        href={Routes.static_path(@socket, "/images/" <> image_file)}
      />
    <% end %>
    ...
    """
  end
end
Lastly - cannonball.svg - we will make this file from scratch!
$ touch priv/static/images/cannonball.svg

<svg width="100" height="100" version="1.1" xmlns="http://www.w3.org/2000/svg">
  <circle cx="50" cy="50" r="25" stroke="black" fill="gray" stroke-width="5" />
</svg>
The width and height values will be overriden by our LiveView render's width="1" height="1", but they still play an important role - because the circle's parameters will be measured relative to these - so we'll set them to 100 for simplicity.  cx and cy represent the coordinates for the center of the circle, which should be one-half the width and height.  The size of the circle will be set with r (radius) and stroke-width (the border around the circle) - we can calculate diameter = 2 * (r + stroke_width) = 60.  This diameter is also relative, so when our cannonball.svg is scaled down to 1 x 1, the visible circle will be 0.6 x 0.6

ECSx is an Entity-Component-System (ECS) framework for Elixir.
In ECS:
	Every game object is an Entity, represented by a unique ID.
	The data which comprises an Entity is split among many Components.
	Game logic is split into Systems, which update the Components every server tick.

Under the hood, ECSx uses Erlang Term Storage (ETS) to store active Components in memory.
A single GenServer manages the ETS tables to ensure strict serializability and customize
the run order for Systems.

      Link to this function

          @spec manager() :: module() | nil


      


Returns the ECSx manager module.
This is set in your app configuration:
config :ecsx, manager: MyApp.Manager

          @spec manager_path() :: binary() | nil


      


Returns the path to the ECSx manager file.
This is inferred by your module name.  If you want to rename or move the
manager file so the path and module name are no longer in alignment, use
a custom :path opt along with the manager module, wrapped in a tuple.

  
  examples

  
  Examples


# standard path: lib/my_app/manager.ex
config :ecsx, manager: MyApp.Manager

# custom path: lib/foo/bar/baz.ex
config :ecsx, manager: {MyApp.Manager, path: "lib/foo/bar/baz.ex"}

          @spec tick_rate() :: integer()


      


Returns the tick rate of the ECSx application.
This defaults to 20, and can be changed in your app configuration:
config :ecsx, tick_rate: 15

A store to which clients can write, for communication with the ECSx backend.
Events are created from the client process by calling add/2, then retrieved by the handler
system using get_and_clear/0.  You will be required to create the handler system yourself -
see the tutorial project for a detailed example.

          @spec add(id(), any()) :: :ok


      


Add a new client event.
The first argument is the entity which spawned the event.
The second argument can be any representation of the event, usually either an atom or a tuple
containing an atom name along with additional metadata.

  
  examples

  
  Examples


# Simple event requiring no metadata
ECSx.ClientEvents.add(player_id, :spawn_player)

# Event with metadata
ECSx.ClientEvents.add(player_id, {:send_message_to, recipient_id, message})

Returns a specification to start this module under a supervisor.
See Supervisor.

  



  
    
      
      Link to this function
    
    get_and_clear()


      
       
       View Source

Returns the list of events, simultaneously clearing it.
This function guarantees that each event is returned exactly once.

A Component labels an entity as having a certain attribute, and holds any data needed to model that attribute.
For example, if Entities in your application should have a "color" value, you will create a Component type called Color.  This allows you to add a color component to an Entity with add/2, look up the color value for a given Entity with get_one/1, get all Entities' color values with get_all/1, remove the color value from an Entity altogether with remove/1, or test whether an entity has a color with exists?/1.
Under the hood, we use ETS to store the Components in memory for quick retrieval via Entity ID.
Usage
Each Component type should have its own module, where it can be optionally configured.
defmodule MyApp.Components.Color do
  use ECSx.Component,
    value: :binary,
    unique: true
end
Options
	:value - The type of value which will be stored in this component type.  Valid types are: :atom, :binary, :datetime, :float, :integer
	:unique - When true, each entity may have, at most, one component of this type;  attempting to add another will overwrite the first.  When false, an entity may have many components of this type.
	:read_concurrency - when true, enables read concurrency for this component table.  Only set this if you know what you're doing.  Defaults to false

          @callback add(entity :: id(), value :: value()) :: :ok


      


Creates a new component.

  Example


# Add an ArmorRating component to entity `123` with value `10`
ArmorRating.add(123, 10)

Look up all components of this type.

# Get all velocity components
Velocity.get_all()

Look up all components of this type belonging to a given entity.
This function is only useful for component types configured with unique: false.
For unique components, get_one/1 should be used instead.

# Get all PowerUp components for entity `123`
PowerUp.get_all(123)

          @callback get_one(entity :: id()) :: value() | nil


      


Look up a single component.
Raises if more than one component is returned.

# Get the Velocity for entity `123`
Velocity.get_one(123)

  



  
    
      
      Link to this callback
    
    remove(entity)


      
       
       View Source
     


  


  

      

          @callback remove(entity :: id()) :: :ok


      


Removes any existing components of this type from an entity.

  



  
    
      
      Link to this callback
    
    remove_one(entity, value)


      
       
       View Source
     


  


  

      

          @callback remove_one(entity :: id(), value :: value()) :: :ok


      


Removes one component with a specific entity ID and value.
This function is only useful for component types configured with unique: false.
For unique components, remove/1 should be used instead.

  
  example

  
  Example


# Remove a specific PowerUp value `9` from entity `123`
PowerUp.remove_one(123, 9)

  



  
    
      
      Link to this callback
    
    search(value)


      
       
       View Source
     


  


  

      

          @callback search(value :: value()) :: [id()]


      


Look up all IDs for entities which have a component of this type with a given value.

  
  example

  
  Example


# Get all entities with a velocity of `60`
Velocity.search(60)

  


        

      



  

    
ECSx.Manager 
    



      
The Manager for your ECSx application.
In an ECSx application, the Manager is responsible for:
	starting up ETS tables for each Component Type, where the Components will be stored
	prepopulating the game content into memory
	keeping track of the Systems to run, and their run order
	running the Systems every tick

components/0 and systems/0
Your Manager module must contain two zero-arity functions called components and systems
which return a list of all Component Types and Systems in your application.  The order of
the Component Types list is irrelevant, but the order of the Systems list is very important,
because the Systems are run consecutively in the given order.
setup block
Another important piece of the Manager module is the setup block.  Here you can load
all the necessary data for your app before any Systems run or users connect.  See setup/1
for more information.

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Functions
  


    
      
        setup(block)

      


        Runs the given code block during startup.



    





      


      
        
          
            
            Anchor for this section
          
Functions
        

        


  
    
      
      Link to this macro
    
    setup(block)


      
       
       View Source
     


      (macro)

  


  

Runs the given code block during startup.
The code will be run during the Manager's initialization (so pay special attention to the
position of ECSx.Manager in your application's supervision tree). The Component tables will
be created before setup is executed.

  
  example

  
  Example


defmodule YourApp.Manager do
  use ECSx.Manager

  setup do
    for npc <- YourApp.fetch_npc_spawn_info() do
      YourApp.Components.Name.add(npc.id, npc.name)
      YourApp.Components.HitPoints.add(npc.id, npc.hp)
      YourApp.Components.Location.add(npc.id, npc.spawn_location)
    end
  end
end
This setup will spawn each NPC with Components for Name, HitPoints, and Location.

  


        

      



  

    
ECSx.System behaviour
    



      
A fragment of game logic which reads and updates Components.
Every System must implement a run function.
By default, the System will run every game tick.  To use a longer period between runs,
you can pass the option :period.  For example, to set a System to run every 5 ticks:
use ECSx.System,
  period: 5

      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Callbacks
  


    
      
        run()

      


        Invoked to run System logic.



    





      


      
        
          
            
            Anchor for this section
          
Callbacks
        

        


  
    
      
      Link to this callback
    
    run()


      
       
       View Source
     


  


  

      

          @callback run() :: :ok


      


Invoked to run System logic.
This function will be called every T game ticks, where T is the value of
the System's :period option (defaults to 1).
Note:  A crash inside this function will restart the entire app!

  


        

      



  

    
ECSx.Tag behaviour
    



      
A component type which does not require a value.  This is useful when the mere presence or absence of a component is all the information we need.
For example, if we want a component type to model a boolean attribute, such as whether or not players may target a particular entity, we'll use a Tag:
defmodule MyApp.Components.Targetable do
  use ECSx.Tag
end
Then we can check for targetability with ...Targetable.exists?(entity) or get a list of all targetable entities with ...Targetable.get_all().
Options
	:read_concurrency - when true, enables read concurrency for this component table.  Only set this if you know what you're doing.  Defaults to false


      


      
        
          
            
            Anchor for this section
          
          Summary
        


  
    Types
  


    
      
        id()

      


    





  
    Callbacks
  


    
      
        add(entity)

      


        Creates a new tag for a given entity.



    


    
      
        exists?(entity)

      


        Checks if an entity has this tag.



    


    
      
        get_all()

      


        Gets a list of all entities with this tag.



    


    
      
        remove(entity)

      


        Removes this component from an entity.

Types
        

        


  
    
      
      Link to this type
    
    id()


      
       
       View Source
     


  


  

      

          @type id() :: any()

      Link to this callback
    
    add(entity)


      
       
       View Source
     


  


  

      

          @callback add(entity :: id()) :: :ok


      


Creates a new tag for a given entity.

  



  
    
      
      Link to this callback
    
    exists?(entity)


      
       
       View Source
     


  


  

      

          @callback exists?(entity :: id()) :: boolean()


      


Checks if an entity has this tag.

  



  
    
      
      Link to this callback
    
    get_all()


      
       
       View Source
     


  


  

      

          @callback get_all() :: [id()]


      


Gets a list of all entities with this tag.

  



  
    
      
      Link to this callback
    
    remove(entity)


      
       
       View Source
     


  


  

      

          @callback remove(entity :: id()) :: :ok


      


Removes this component from an entity.

  


        

      



  

    
ECSx.MultipleResultsError exception
    







  

    
mix ecsx.gen.component 
    



      
Generates a new Component type for an ECSx application.
$ mix ecsx.gen.component Height integer

The first argument is the name of the component, followed by the data type of the value.
Valid types for the component's value are:
	atom
	binary
	datetime
	float
	integer

By default, new component types are generated with unique: true, which allows an entity to have at most one component of this type at any given time.  To override this, use the --no-unique flag:
$ mix ecsx.gen.component Friendship binary --no-unique


      





  

    
mix ecsx.gen.system 
    



      
Generates a new System for an ECSx application.
$ mix ecsx.gen.system Foo

The only argument accepted is a module name for the System.

      





  

    
mix ecsx.gen.tag 
    



      
Generates a new ECSx Tag - a Component type which doesn't store any value.
$ mix ecsx.gen.tag Attackable

The single argument is the name of the component.

      





  

    
mix ecsx.setup 
    



      
Generates the Manager process which runs an ECSx application.
$ mix ecsx.setup

This setup will generate manager.ex and empty folders for components and systems.
If you don't want to generate the folders, you can provide option --no-folders

